

```

//*****
/* MACRO: TEXTBTTN.WCM
/* PURPOSE: Create text on a grey button.
//*****
GLOBAL HRt = NTOC(0F90Ah)
Application (Presentations; "Presentations"; Default; "EN")
if((EnvCurrentMode != 1)AND(EnvCurrentMode != 2))
    MESSAGEBOX(vStatus; "Text Button Error"; "Presentations must be in slide editor or drawing mode."; IconStop!)
    Quit
endif
DialogDefine(200; 50; 50; 230; 65;Percent!|Cancel!; "Text Button Macro")
DialogAddText(200; 101; 10; 5; 75; 13; Left!; "&Button Text:")
DialogAddEditBox(200; 102; 10; 16; 150; 12; Left!|WPChars!; vText; 50)
DialogAddText(200; 103; 10; 37; 150; 13; Left!; "Press Ctrl+W to insert symbols or characters.")
DialogAddPushButton(200; "OKButton"; 175; 10; 45; 13; 1; "OK")
DialogAddPushButton(200; "CANCELButton"; 175; 28; 45; 13; 0; "Cancel")
DialogLoad(200; "Presentations")
RegionShowWindow("200.CancelBttn"; Hide!)
DialogShow(200; 1; Callback@)
CallbackWait ()
DialogDestroy(200)
    BeginBulletChart ()
    ChartSetCreateSize (5195; 5860; 9100; 5428)
    ChartCreateBullet ()
    ModifyObjects ()
    Type (vText)
    BulletChartAttrStart ()
    for(x; 1; x<7; x+1)
        SetBulletDefinition (x; "")
    endfor
    SetBulletLayout (63; Center!; Left!; Left!; Left!; Left!)
    SetTextChartBox (All!; Behind!)
    SetBorder (TextChart!; Rectangle!; Bevel!; 0; 0; 0)
    SetGradient (BoxTitle!; 0; 192; 192; 192; 255; 255; 255; 2; 0; 0; 0; 64; On!)
    SetPattern (TextChart!; Solid!; 192; 192; 192; 255; 255; 255; None!; On!)
    BulletChartAttrEnd (Chart!)
    TextEditExit ()
QUIT

LABEL(Callback)
if(Callback[5]=274)
    DialogDestroy(200)
    QUIT
endif
if(Callback[5]=273)
    switch(Callback[3])
        caseof"OKButton":
            DialogDismiss("200"; "OKButton")
            CallbackResume ()
        caseof"CancelBttn":
            DialogDestroy(200)
            QUIT
        caseof"CANCELButton":
            DialogDestroy(200)
            QUIT
    endswitch
endif
RETURN

```